Illustrating the Continuous Wavelet Transform (CWT) (Mathematica 10)

Definitions and conventions

CWT and Discrete Approximation

- The continuous wavelet transform of a function $f(t)$ is given by

$$w(s, a) = \frac{1}{\sqrt{s}} \int_{-\infty}^{\infty} f(t)\, \psi^*\left(\frac{t-a}{s}\right) dt.$$

- The continuous wavelet transform of a uniformly sampled sequence $\{x_1, x_2, \ldots\} = \{x(t_0), x(\Delta + t_0), \ldots\}$ is given by

$$w(s, a) = \frac{1}{\sqrt{s}} \sum_{k=1}^{n} x_k\, \psi^*\left(\frac{\Delta\,(k-a)}{s}\right)$$

Mathematica Input

`ContinuousWaveletTransform[{x1, x2, ...}]`
gives the continuous wavelet transform of a list of values x_i.

`ContinuousWaveletTransform[data, wave]`
gives the continuous wavelet transform using the wavelet *wave*.

`ContinuousWaveletTransform[data, wave, {noct, nvoc}]`
gives the continuous wavelet transform using *noct* octaves with *nvoc* voices per octave.

`ContinuousWaveletTransform[sound, ...]`
gives the continuous wavelet transform of sampled sound.

Scaling details

- The scaling parameter s is given by equal-tempered scale $s_{\text{oct,voc}} = \alpha\, 2^{\text{oct}-1}\, 2^{\text{voc/nvoc}}$ where oct is the octave number, voc the voice number, and α the smallest wavelet scale.
- The default value for `noct` is given by $\left\lfloor \log_2\left(\frac{n}{2}\right) \right\rfloor$, where n is the length of the input.
- The default value for `nvoc` is 4.
- The scaling parameter s is given by equal-tempered scale $s_{\text{oct,voc}} = \alpha\, 2^{\text{oct}-1}\, 2^{\text{voc/nvoc}}$ where oct is the octave number, voc the voice number, and α the smallest wavelet scale.
- For each scale $s_{\text{oct,voc}}$, the `ContinuousWaveletTransform` computes the wavelet coefficients $\{w(1, s_{\text{oct,voc}}), w(2, s_{\text{oct,voc}}), \ldots, w(n, s_{\text{oct,voc}})\}$.

CWT in Mathematica

- `MorletWavelet[...]`

The wavelet function (ψ) is given by $1 \Big/ \sqrt[4]{\pi}\, \cos\left(x\, \pi \sqrt{2/\log(2)}\right) \exp\left(-\frac{x^2}{2}\right)$.

In[136]:=
```
Plot[WaveletPsi[MorletWavelet[], x], {x, -5, 5}, PlotRange → All, Filling → Axis]
```

Out[136]=



In[137]:=
```
Plot[WaveletPsi[MorletWavelet[], (x - 1) / 2] / Sqrt[2],
 {x, -5, 5}, PlotRange → All, Filling → Axis]
```

Out[137]=



In[138]:=
```
Plot[WaveletPsi[MorletWavelet[], 2 (x + 1)] Sqrt[2],
 {x, -5, 5}, PlotRange → All, Filling → Axis]
```

Out[138]=



■ `GaborWavelet[...]`

The wavelet function (ψ) is given by $\frac{1}{\sqrt[4]{\pi}} \exp(i w x) \exp\left(-\frac{x^2}{2}\right)$.

In[139]:=
```
ψ = WaveletPsi[GaborWavelet[1], x];
```

In[140]:=
```
Plot[{Re[ψ], Im[ψ]}, {x, -5, 5},
 PlotRange → All, PlotStyle → {Red, Blue}, Filling → Axis]
```

Out[140]=



■ `DGaussianWavelet[...]`

The wavelet function (ψ) is given by $\frac{(-1)^{n+1}}{\sqrt{\Gamma\left(n+\frac{1}{2}\right)}} \frac{\partial^n}{\partial x^n} \exp\left(-\frac{x^2}{2}\right)$.

In[141]:=
```
Plot[WaveletPsi[DGaussianWavelet[1], x], {x, -5, 5}, Filling → Axis]
```

Out[141]=



In[142]:=
```
Plot[WaveletPsi[DGaussianWavelet[10], x], {x, -5, 5}, Filling → Axis]
```

Out[142]=



■ `MexicanHatWavelet[...]` (default)

The wavelet function (ψ) is given by $\frac{-2}{\sqrt[4]{\pi}\sqrt{3\sigma}}\left(\frac{x^2}{\sigma^2}-1\right)\exp\left(-\frac{x^2}{2\sigma^2}\right)$.

In[143]:=
```
Plot[WaveletPsi[MexicanHatWavelet[1], x],
 {x, -5, 5}, PlotRange → All, Filling → Axis]
```

Out[143]=



- `PaulWavelet[...]`

The wavelet function (ψ) is given by $\frac{(2i)^n n!}{\sqrt{\pi}\sqrt{(2n)!}}(1 - i x)^{-n-1}$.

In[144]:=
```
ψ = WaveletPsi[PaulWavelet[4], x];
```

In[145]:=
```
Plot[{Re[ψ], Im[ψ]}, {x, -2, 2},
 PlotRange → All, PlotStyle → {Red, Blue}, Filling → Axis]
```

Out[145]=



Inverse CWT

`InverseContinuousWaveletTransform[cwd]`
gives the inverse continuous wavelet transform of a `ContinuousWaveletData` object *cwd*.

`InverseContinuousWaveletTransform[cwd, wave]`
gives the inverse transform using the wavelet *wave*.

`InverseContinuousWaveletTransform[cwd, wave, octvoc]`
gives the inverse transform from the wavelet coefficients specified by *octvoc*.

Scalogram

`WaveletScalogram[wd]`
plots wavelet vector coefficients in a `DiscreteWaveletData` or `ContinuousWaveletData` object *wd*.

`WaveletScalogram[wd, wind]`
plots wavelet coefficients corresponding to the wavelet index specification *wind*.

`WaveletScalogram[wd, wind, func]`
applies *func* to coefficients before plotting.

Examples

Example 1

In[146]:=
```
cwt1 = ContinuousWaveletTransform[{56, 40, 8, 24, 48, 48, 40, 16}]
```

Out[146]=
```
ContinuousWaveletData[  Octaves: 2  Voices: 4
                                                                          Data points: 8 ]
```

In[147]:=
```
Normal[cwt1]
```

Out[147]=
```
{{1, 1} → {7.49524, 1.43189, -5.46784,
    -0.991773, 3.14778, 0.161406, 2.93329, -8.70999}, {1, 2} →
  {10.4739, 2.38737, -8.16256, -1.43732, 4.43939, 0.726138, 3.66238, -12.0893},
 {1, 3} → {14.1227, 3.90981, -11.943, -2.10298,
   6.10712, 1.91918, 4.10948, -16.1223}, {1, 4} →
  {18.1873, 6.16541, -16.932, -3.16789, 8.18809, 4.13231, 3.81324, -20.3865},
 {2, 1} → {22.1232, 9.11941, -22.895, -4.97487,
   10.7244, 7.72386, 2.22932, -24.0503}, {2, 2} →
  {25.137, 12.2426, -28.9579, -8.00357, 13.7414, 12.6935, -0.886958, -25.966},
 {2, 3} → {26.3599, 14.3123, -33.597, -12.4843,
   17.019, 18.2768, -4.83573, -25.051}, {2, 4} →
  {25.0448, 13.733, -35.3052, -17.5626, 19.5819, 22.8807, -7.55786, -20.8147}}
```

In[148]:=
```
ListLinePlot[cwt1[All, "Values"]]
```

Out[148]=



In[149]:=
```
WaveletScalogram[cwt1, AxesLabel → {"time", "scale"}]
```

Out[149]=



In[150]:=
```
InverseContinuousWaveletTransform[cwt1]
```

Out[150]=
```
{56., 40., 8., 24., 48., 48., 40., 16.}
```

Example 2

In[151]:=
```
cwt2 = ContinuousWaveletTransform[{0, 0, 1, 0, 0}, MexicanHatWavelet[1]]
```

Out[151]=
```
ContinuousWaveletData[ ⊞  Octaves: 1  Voices: 4
Data points: 5 ]
```

In[152]:=
```
Normal[cwt2]
```

Out[152]=
```
{{1, 1} → {0.0119594, -0.144708, 0.265497, -0.144708, 0.0119594},
 {1, 2} → {0.0100158, -0.196057, 0.372082, -0.196057, 0.0100158},
 {1, 3} → {-0.0000441977, -0.251178, 0.502445, -0.251178, -0.0000441977},
 {1, 4} → {-0.0263966, -0.296391, 0.645576, -0.296391, -0.0263966}}
```

In[153]:=
```
WaveletScalogram[cwt2, AxesLabel → {"time", "scale"}]
```

Out[153]=



In[154]:=
```
InverseContinuousWaveletTransform[cwt2]
```

Out[154]=
```
{-1.08247 × 10^-15, -5.55112 × 10^-16, 1., -2.77556 × 10^-16, -9.71445 × 10^-16}
```

Example 3

In[155]:=
```
cwt3 = ContinuousWaveletTransform[Sin[Range[40] / 5]]
```

Out[155]=
```
ContinuousWaveletData[ <image> Octaves: 4  Voices: 4
                       Data points: 40 ]
```

In[156]:=
```
cwt3["IndexMap"]
```

Out[156]=
```
{{1, 1}, {1, 2}, {1, 3}, {1, 4}, {2, 1}, {2, 2}, {2, 3}, {2, 4},
 {3, 1}, {3, 2}, {3, 3}, {3, 4}, {4, 1}, {4, 2}, {4, 3}, {4, 4}}
```

In[157]:=
```
WaveletScalogram[cwt3, Ticks → Full]
```

Out[157]=



Example 4

In[158]:=
```
data4 = Table[Sign[Cos[x^2]], {x, -6, 6, 12. / 1023}];
```

In[159]:=
```
ListLinePlot[data4]
```

Out[159]=



In[160]:=
```
cwt4 = ContinuousWaveletTransform[data4];
```

In[161]:=
```
WaveletScalogram[cwt4, AxesLabel → {"time", "scale"}]
```

Out[161]=



In[162]:=
```
ListPlot3D[cwt4[All, "Values"], ColorFunction → "DarkRainbow",
 AxesLabel → {"time", "scale", " "}, Mesh → None, PlotRange → All]
```



Parameters and resolution

Voices per octave

Example 5-6

```
data56 = Table[Sin[x^2], {x, 0, 10, 0.02}];

{cwt5, cwt6} = Table[ContinuousWaveletTransform[
    data56, DGaussianWavelet[5], {Automatic, voc}], {voc, {4, 16}}];
```

```
{WaveletScalogram[cwt5, PlotLabel → "4 voices/octave"],
 WaveletScalogram[cwt6, PlotLabel → "16 voices/octave"]}
```



Number of octaves

Example 7-8

```
data78 = Function[x, x Sin[Exp[x] x]][Range[2, 5, 0.001]];

{cwt7, cwt8} = Table[ContinuousWaveletTransform[
    data78, Automatic, {oct, Automatic}], {oct, {5, 8}}];

{WaveletScalogram[cwt7, PlotLabel → "5 octaves"],
 WaveletScalogram[cwt8, PlotLabel → "8 octaves"]}
```



Localization

Harmonic wave

Example 9

```
data9 = Table[Cos[20 π x], {x, 0, 1, 1/1023}];
```

```
ListLinePlot[data9]
```




```
cwt9 = ContinuousWaveletTransform[data9, GaborWavelet[6]];

WaveletScalogram[cwt9, ColorFunction → "SolarColors"]
```



Superposition of harmonics

Example 10

```
data10 = Table[Cos[20 π x] + Cos[√2 50 π x], {x, 0, 1, 1/1023}];

ListLinePlot[data10]
```



```
cwt10 = ContinuousWaveletTransform[data10, GaborWavelet[6]];

WaveletScalogram[cwt10, ColorFunction → "SolarColors"]
```



Variable frequencies

Example 11

```
data11 = Block[{fr = 0}, Table[Sin[(fr += 0.5) x], {x, 0, 1, 1/1023}]];

ListLinePlot[data11]
```




```
cwt11 = ContinuousWaveletTransform[data11, GaborWavelet[6]];
```

```
WaveletScalogram[cwt11, ColorFunction → "SolarColors"]
```



Sharply localized events

Example 12

```
data12 = Table[HeavisideTheta[x], {x, -2, 2, 4/1023}];
```

```
ListLinePlot[data12]
```




```
cwt12 = ContinuousWaveletTransform[data12,
  DGaussianWavelet[4], {Automatic, 12}, Padding → "Fixed"]
```

ContinuousWaveletData[Octaves: 10 Voices: 12
Data points: 1024]

```
WaveletScalogram[cwt12]
```



Example 13

```
data13 = Table[Sin[4 π t] + 2 Exp[-10^5 (1/3 - t)^2], {t, 0, 1, 1/511}];
```

```
ListLinePlot[data13]
```




```
cwt13 =
  ContinuousWaveletTransform[data13, PaulWavelet[5], {7, 12}, SampleRate → 511];
```

```
WaveletScalogram[cwt13, ColorFunction → "AvocadoColors"]
```



Example 14

```
data14 = Table[Piecewise[{{Sin[2 π 10 t], 0 ≤ t < 1/4}, {Sin[2 π 25 t], 1/4 ≤ t < 1/2},
    {Sin[2 π 50 t], 1/2 ≤ t < 3/4}, {Sin[2 π 100 t], 3/4 ≤ t ≤ 1}}], {t, 0, 1, 1/1023}];
ListLinePlot[data14, AspectRatio → 0.5]
```




```
cwt14 = ContinuousWaveletTransform[data14, DGaussianWavelet[6], {Automatic, 12}];
WaveletScalogram[cwt14, ColorFunction → "RustTones"]
```



```
ListPlot3D[cwt14[All, "Values"], ColorFunction → "DarkRainbow",
 AxesLabel → {"time", "scale", "   "}, Mesh → None, PlotRange → All]
```



Example 15

```
data15 = Table[Cos[40 π x] Exp[-100 π (x - 0.7)^2] +
    Cos[140 π x] Exp[-50 π (x - 0.2)^2], {x, 0, 1, 1/1023}];
```

```
ListLinePlot[data15, PlotRange → All]
```




```
cwt15 =
  ContinuousWaveletTransform[data15, MexicanHatWavelet[4], SampleRate → 1023];
```

```
WaveletScalogram[cwt15, ColorFunction → "RustTones"]
```




```
ListPlot3D[Abs@cwt15[All, "Values"],
 ColorFunction → "DarkRainbow",
 AxesLabel → {"time", "scale"},
 Mesh → None, PlotRange → All]
```



Wavelet families for CWT

Example 16

```
data16 = Table[Sin[x^2] + Cos[x^3], {x, 0, 3 π, 0.01}];
```

```
ListLinePlot[data16]
```




```
cwt16mex = ContinuousWaveletTransform[data16, MexicanHatWavelet[1]];
```

```
WaveletScalogram[cwt16mex]
```




```
cwt16gabor = ContinuousWaveletTransform[data16, GaborWavelet[3]]
```

ContinuousWaveletData[Octaves: 8 Voices: 4 Data points: 943]

```
WaveletScalogram[cwt16gabor]
```



Example 17

```
data17 = Table[Sign[Cos[x^2]], {x, -6, 6, 12./1023}];
```

```
ListLinePlot[data17]
```




```
cwt17mex = ContinuousWaveletTransform[data17, MexicanHatWavelet[]]
```

ContinuousWaveletData[Octaves: 9 Voices: 4 Data points: 1024]

```
ListPlot3D[Abs@cwt17mex[All, "Values"],
 AxesLabel → {"time", "scale", " "},
 Mesh → None,
 PlotRange → All]
```




```
cwt17dgauss = ContinuousWaveletTransform[data17, DGaussianWavelet[]];
```

```
ListPlot3D[Abs@cwt17dgauss[All, "Values"],
 ColorFunction → "DarkRainbow",
 AxesLabel → {"time", "scale", " "},
 Mesh → None, PlotRange → All]
```




```
cwt17gabor = ContinuousWaveletTransform[data17, GaborWavelet[3]];
```

```
ListPlot3D[Abs@cwt17gabor[All, "Values"],
 ColorFunction → "DarkRainbow",
 AxesLabel → {"time", "scale", " "},
 Mesh → None, PlotRange → All]
```




```
cwt17paul = ContinuousWaveletTransform[data17, PaulWavelet[]];
```

```
ListPlot3D[Abs@cwt17paul[All, "Values"],
 ColorFunction → "DarkRainbow",
 AxesLabel → {"time", "scale"},
 Mesh → None, PlotRange → All]
```



Complex CWT for amplitude and phase

Example 18

```
f[t_, ϵ_] := √(t (1 - t)) Sin[2 π (1 + ϵ) / (t + ϵ)]
```

```
data18 = Table[f[t, 0.05], {t, 0, 1, 1/1023}];
```

```
ListLinePlot[data18]
```




```
cwt18 = ContinuousWaveletTransform[data18, GaborWavelet[6], {Automatic, 16}]
```

ContinuousWaveletData[Octaves: 9 Voices: 16 Data points: 1024]

```
WaveletScalogram[cwt18, ColorFunction → "BlueGreenYellow"]
```




```
WaveletScalogram[cwt18, Automatic, Arg, ColorFunction → "BlueGreenYellow"]
```



Filtering by using CWT

Example 19

```
f[x_] := Sin[50 π x] Exp[-100 π (x - 0.5)^2] +
   (Sin[50 π x] + 2 Cos[140 π x]) Exp[-50 π (x - 0.2)^2] +
   2 Sin[150 π x] Exp[-80 π (x - 0.8)^2]

data19 = Table[f[x], {x, 0., 1, 1/2047}];
```

```
ListLinePlot[data19, PlotRange → All]
```




```
cwt19 = ContinuousWaveletTransform[data19, GaborWavelet[6],
    {10, 16}, Padding → 0.0, SampleRate → 2047, WaveletScale → Automatic];

WaveletScalogram[cwt19, ColorFunction → "SolarColors"]
```




```
vec = ConstantArray[1, Length[data19]];
vec[[1 ;; 1000]] := 0.0

cwt19Thresh = WaveletMapIndexed[#1 vec &, cwt19, {{4 | 5, _}, {6, u_ /; u < 8}}];

WaveletScalogram[cwt19Thresh, ColorFunction → "SolarColors"]
```




```
rdata19 = InverseContinuousWaveletTransform[cwt19Thresh];
```

```
ListLinePlot[{data19[[1 ;; 800]], rdata19[[1 ;; 800]]},
 PlotRange → All, PlotStyle → {Blue, {Red, Thick}}]
```




```
ListLinePlot[rdata19, PlotRange → All]
```



Example 20

```
data20 = Abs[Zeta[1/2 + i Range[0, 100, 0.05]]];
```

```
ListLinePlot[data20]
```



```
cwt20 = ContinuousWaveletTransform[
  data20, GaborWavelet[6], {4, 12}, WaveletScale → 80]
```

```
ContinuousWaveletData[ ⊞ Octaves: 4  Voices: 12
                       Data points: 2001 ]
```

```
WaveletScalogram[cwt20, Automatic, Re, ColorFunction → "SolarColors"]
```




```
WaveletScalogram[cwt20, Automatic, Abs, ColorFunction → "BlueGreenYellow"]
```




```
the end
```